

## CÁC HÀM THÔNG DỤNG & CÔNG THỨC MẢNG

### ➡ Nhóm hàm số học:

- **Hàm INT:** Hàm lấy giá trị là phần nguyên.  
Cú pháp: **INT (X)**  
Hàm cho giá trị là phần nguyên của X  
VD: =INT(10/3) =INT(3.9); =INT(3); INT(3.2) đều cho kết quả là 3
- **Hàm MOD:** Hàm cho giá trị là phần dư.  
Cú pháp: **MOD ( a,b)**  
Hàm MOD cho giá trị là phần dư của phép chia a:b  
VD: =MOD(16,3) Kết quả {1}
- **Hàm ROUND:** Hàm làm tròn số.

Cú pháp: **ROUND ( X,n)**

Hàm làm tròn n số của X

- Nếu n dương ( n>0) sẽ làm tròn số bên phải kể từ vị trí dấu chấm thập phân

=	=ROUND(1234.567,2)
A	B
KQ	1234.57

- Nếu n âm ( n<0) sẽ làm tròn số bên trái kể từ vị trí dấu chấm thập phân.

=	=ROUND(1234.567,-2)
A	B
KQ	1200

### ➡ Nhóm hàm ngày tháng:

- **Hàm TODAY:** Hàm cho giá trị là ngày tháng hiện tại của máy tính.

Cú pháp: **TODAY()**

<i>f<sub>x</sub></i>	=TODAY()
C	D
KQ	20/06/2008

- **Hàm DAY:** Hàm cho giá trị là ngày của biểu thức số.

Cú pháp: **DAY(số)**

=DAY(D1)
D
20/06/2008
20

- **Hàm MONTH:** Hàm cho giá trị là tháng của biểu thức số.

Cú pháp: **MONTH ( số)**

=MONTH(D1)
D
20/06/2008
6

- **Hàm YEAR:** Hàm cho giá trị là năm của biểu thức số.  
Cú pháp: **YEAR (số)**

- **Hàm DAYS360:** Hàm cho giá trị là số ngày giữa khoảng cách của BTNT1 và BTNT2.

Cú pháp: **DAYS360 ( BTNT1, BTNT2)**

=DAYS360(D1,D2)	
D	E
20/06/2008	
22/06/2008	2

### Nhóm hàm văn bản

- **Hàm LEFT:** Hàm lấy ký tự từ bên trái sang

Cú pháp: **LEFT ( X,n)** Hàm LEFT lấy n ký tự từ bên trái sang của văn bản X

=LEFT(B1,3)	
B	C
Mai Hoa	Mai

- **Hàm RIGHT:** Hàm lấy ký tự từ bên phải sang

Cú pháp: **RIGHT ( X,n)** Hàm RIGHT lấy n ký tự từ bên phải sang của văn bản X

=	=RIGHT(B1,3)
B	C
Mai Hoa	Hoa

- **Hàm MID:** Hàm lấy ký tự bên trong của chuỗi.

Cú pháp: **MID (X, m, n)** Hàm MID lấy n ký tự trong chuỗi X bắt đầu từ vị trí m.

=	=MID(B1,8,3)
B	C
Nguyễn Hoa Mai	Hoa

### ➤ Nhóm hàm Logic

- **Hàm AND:** và

Cú pháp: **AND ( điều kiện 1, điều kiện 2, ...)** Hàm nhận giá trị đúng nếu mọi điều kiện là đúng, ngược lại hàm nhận giá trị sai

Ví dụ : = AND (5>3,2>0,-2>-8) Kết quả {True}

- **Hàm OR:** hoặc

Cú pháp: **OR ( điều kiện 1, điều kiện 2, ....)** Hàm nhận giá trị đúng nếu tồn tại một điều kiện là đúng, ngược lại hàm nhận giá trị sai

Ví dụ : = OR (3>3,2>8,4>8,1>0) Kết quả {True}

- **Hàm IF:** Nếu

Cú pháp: **IF ( điều kiện, giá trị 1, giá trị 2)** Nếu điều kiện đúng thì hàm trả về giá trị 1, ngược lại hàm nhận giá trị 2

VD 1: Tính PCCV, biết rằng nếu chức vụ là Giám đốc thì PCCV: 150000, ngoài ra không có PCCV.

Tại ô C2 ta đánh công thức:

=IF(B2="GD",150000,0)		
	B	C
	Chức vụ	PCCV
	GD	150000
	TP	0
	PP	0
	NV	0

VD 2 Tính tiền Khen thưởng, biết rằng:

- Nếu chức vụ là NV và Mã KT là A, thì được thưởng 150000 đ. Ngoài ra không có thưởng.

Tại ô D2 ta đánh công thức:

=IF(AND(B2="NV",C2="A"),150000,0)			
	B	C	D
	Chức vụ	Mã KT	Tiền thưởng
	TP	A	0
	NV	B	0
	NV	A	150000

VD3: Tính tiền Thưởng, biết rằng:

Nếu chức vụ là NV hoặc Mã KT là A, thì được thưởng 150000 đ. Ngoài ra không có thưởng.

Tại ô D2 ta đánh công thức:

=IF(OR(B2="NV",C2="A"),150000,0)			
	B	C	D
	Chức vụ	Mã KT	Tiền thưởng
	NV	A	150000
	PP	A	150000
	PP	B	0

Nếu có nhiều điều kiện ( nhiều hàm IF ) thì cuối dãy hàm IF đó ta mới đóng ngoặc. Nghĩa là có bao nhiêu hàm IF thì đóng bấy nhiêu ngoặc.

VD: Tính PCCV biết rằng:

- Nếu chức vụ là GD, thì PCCV là: 200.000 đ
- Nếu chức vụ là PGĐ, thì PCCV là: 150.000 đ
- Nếu chức vụ là TP, thì PCCV là: 100.000 đ
- Nếu chức vụ là PP, thì PCCV là: 60.000 đ
- Ngoài ra không có PCCV.

➔ **Nhóm hàm thống kê**

- **Hàm SUM:** Hàm tính tổng

Cú pháp: **SUM (vùng)**

=SUM(B1:B3)	
	<b>B</b>
	20
	35
	50
	<b>105</b>

- **Hàm AVERAGE:** Hàm tính giá trị trung bình

Cú pháp: **AVERAGE (vùng)**

=AVERAGE(B1:B3)	
	<b>B</b>
	20
	35
	50
	<b>35</b>

- **Hàm MAX:** Hàm tìm giá trị lớn nhất.

Cú pháp: **MAX (vùng)**

=MAX(B1:B3)	
	<b>B</b>
	20
	35
	50
	<b>50</b>

- **Hàm MIN:** Hàm tìm giá trị bé nhất.

=MIN(B1:B3)	
<b>B</b>	
	20
	35
	50
	20

Cú pháp: **MIN (vùng)**

- **Hàm RANK:** Hàm xếp thứ hạng.

Cú pháp: **RANK ( ô cần xếp thứ, vùng cần so sánh)**

\*) Vùng cần so sánh:

Thường để ở chế độ giá trị tuyệt đối:\$

VD: Xếp thứ cho bảng dưới đây dựa vào cột tổng điểm:

Tại ô C2 ta đánh công thức

=RANK(B2,\$B\$2:\$B\$4)	
<b>B</b>	<b>C</b>
Tổng điểm	Xếp thứ
10	3
18	1
14	2

- **Hàm COUNT:** Hàm đếm (đếm số ô không trống)

=COUNT(B1:B5)	
<b>B</b>	
Lương chính	
700000	
650000	
650000	
500000	
	4

Cú pháp:

**COUNT ( vùng)**

Hàm COUNT đếm số ô có trong vùng. Nhưng chỉ đếm những ô có kiểu dữ liệu là kiểu số.

- **Hàm COUNTIF:** Hàm đếm có điều kiện

Cú pháp:

**COUNTIF ( Vùng cần đếm, ô điều kiện)**

Đếm số ô có trong vùng với điều kiện đã chỉ ra ở ô điều kiện.

Hàm COUNTIF đếm ô có kiểu dữ liệu là kiểu số và cả kiểu ký tự.

**VD:** Đếm xem có bao nhiêu người có mức lương là: 650.000 đ.

Tại ô B6 ta đánh công thức:

=COUNTIF(B1:B5,B3)	
	B
	Lương chính
	700000
	650000
	650000
	500000
	2

- **Hàm SUMIF:** Hàm tính tổng có điều kiện.

Cú pháp:

**SUMIF ( cột1, điều kiện, cột 2)**

\* Điều kiện bao giờ cũng phải bỏ trong dấu “nháy kép”

**VD:** Tính tổng Lương chính của những người có Chức vụ là TP

Tại ô D6 ta đánh công thức:

=SUMIF(B1:B5,"TP",D1:D5)			
	B	C	D
	Chức vụ	Họ tên	Lương chính
	GD	Vinh	1500000
	TP	Tâm	1200000
	TP	Linh	1200000
	NV	Mai	500000
			2400000

## ➤ Nhóm hàm tìm kiếm và tham chiếu

- **Hàm VLOOKUP:** Hàm tìm kiếm và tham chiếu theo cột.

**Cú pháp:** VLOOKUP ( giá trị tìm, vùng cần tìm, cột cần lấy, cách tìm )  
ý nghĩa:

Vùng cần tìm: thường để ở chế độ giá trị tuyệt đối: \$

Cột cần lấy: ở vùng cần tìm.

Cách tìm: Tìm theo 2 giá trị:

0: Cột bên trái của vùng cần tìm **không sắp xếp** (ngầm định)

1: Sắp xếp tăng dần.

**VD:** Tính tiền thưởng của 1 cơ quan sau, dựa vào Bảng khen thưởng. Nghĩa là lấy dữ liệu từ bảng KT đưa lên bảng trên.

Tại ô D2 ta đánh công thức:

Hàm HLOOKUP: Hàm tìm kiếm và tham chiếu theo hàng.

Tương tự VLOOKUP, chỉ đổi hàng thành cột.

VD: Tại ô D2 ta đánh công thức:

- **Hàm HLOOKUP:**

Hàm tìm kiếm và tham chiếu theo hàng.

Tương tự VLOOKUP, chỉ đổi hàng thành cột.

VD: Tại ô D2 ta đánh công thức:

=HLOOKUP(B2,\$B\$8:\$D\$9,2,0)		
B	C	D
Mã KT	Họ tên	Thưởng
A	Vinh	200000
C	Tâm	100000
B	Linh	150000
A	Mai	200000
Bảng khen thưởng		
A	B	C
200000	150000	100000

## ▣ Các hàm thống kê trong CSDL

- **Hàm DAVERAGE:**

Cú pháp:

**DAVERAGE** ( vùng dữ liệu, cột cần tính, vùng tiêu chuẩn).

ý nghĩa: Hàm tính trung bình các giá trị trong cột thoả mãn điều kiện của vùng tiêu chuẩn trong vùng dữ liệu.

VD: Tính tiền Lương trung bình của những người có chức vụ là NV

Tại ô D6 ta đánh công thức:

=DAVERAGE(B1:D5,3,C7:C8)			
B	C	D	
Chức vụ	Thưởng	Lương	
TP	250000	800000	
PP	200000	700000	
NV	100000	450000	
NV	80000	500000	
		475000	
	Chức vụ		
	NV		



- **Hàm DCOUNT:**

\* Cú pháp:

**DCOUNT(vùng dữ liệu, cột cần đếm, vùng tiêu chuẩn).**

\* ý nghĩa: Đếm số ô chứa lượng giá trị số trong cột thoả mãn đk trong vùng tiêu chuẩn tìm kiếm trong vùng dữ liệu.

VD: Đếm xem có bao nhiêu người có mức lương là 500000.

Tại ô D6 ta đánh công thức:

= =DCOUNT(B1:D5,3,C7:C8)		
B	C	D
Chức vụ	Thưởng	Lương
TP	250000	800000
PP	200000	700000
NV	100000	500000
NV	80000	500000
		2
	Lương	
	500000	

- **Hàm DMAX:**

\* Cú pháp:

**DMAX (vùng dữ liệu, cột giá trị, vùng tiêu chuẩn).**

\* ý nghĩa: Hàm tính giá trị cao nhất trong cột giá trị thoả mãn điều kiện của vùng tiêu chuẩn trong vùng dữ liệu.

VD: Tìm xem những người 26 tuổi ai cao lương nhất.

Tại ô D6 ta đánh công thức:

= =DMAX(C1:D5,2,C7:C8)		
C	D	E
Tuổi	Mức lương	
26	700000	
23	450000	
26	600000	
26	400000	
	700000	
Tuổi		
26		

- **Hàm DMIN:**

\* Cú pháp:

= **DMIN** (vùng dữ liệu, cột giá trị, vùng tiêu chuẩn).

\* ý nghĩa: Hàm tính giá trị nhỏ nhất trong cột giá trị thoả mãn điều kiện của vùng tiêu chuẩn trong vùng dữ liệu.

VD: Tìm xem những người 26 tuổi ai thấp lương nhất.

Tại ô D6 ta đánh công thức:

=		=DMIN(C1:D5,2,C7:C8)	
C	D		
Tuổi	Mức lương		
26	700000		
23	450000		
26	600000		
26	470000		
	470000		
Tuổi			
26			

### • Hàm DSUM:

\* Cú pháp:

**DSUM** (vùng dữ liệu, cột giá trị, vùng tiêu chuẩn).

\* ý nghĩa: Hàm tính tổng trong cột giá trị thoả mãn điều kiện của vùng tiêu chuẩn trong vùng dữ liệu.

VD: Tính tổng tiền lương những người 26 tuổi.  
Tại ô D6 ta đánh công thức:

=		=DSUM(C1:D5,2,C7:C8)	
C	D		
Tuổi	Mức lương		
26	700000		
23	450000		
26	600000		
26	470000		
	1770000		
Tuổi			
26			

## CÔNG THỨC MẢNG

Công thức mảng (Formula Array)

Giả sử cột A là SL, B là ĐG.

Cần tính tổng giá trị (Tổng thành tiền)

Cách làm thông thường

	A	B	C
1	SL	ĐG	
2	4	2	=A2*B2 KQ=8

<b>3</b>	3	3	=A3*B3	KQ=9
<b>4</b>	5	2	=A4*B4	KQ=10
<b>5</b>	TỔNG CỘNG :		=SUM(C1:C3)	KQ=27

Cách tính bằng công thức mảng

	<b>A</b>	<b>B</b>	<b>C</b>
<b>1</b>	<b>SL</b>	<b>ĐG</b>	<b>THÀNH TIỀN</b>
<b>2</b>	4	2	{SUM(A2:A3*B2:B3)} KQ = 27
<b>3</b>	3	3	
<b>4</b>	5	2	
<b>5</b>			

=SUM(A2:A4\*B2:B4)

Nhấn CTRL+SHIFT+ENTER

{=SUM(A2:A4\*B2:B4)}

Giải thích :

{=SUM(A2:A4\*B2:B4)}=SUM(A2\*B2+A3\*B3+A4\*B4)

Theo cách phân tích của Formula Array thì EXCEL sẽ chạy con trỏ lần lượt từ dòng đầu tiên trong vùng dữ liệu A1:B1 đến dòng cuối của vùng là A3:B3.

Mỗi một dòng khi con trỏ chạy qua Excel thực hiện công thức  $A2*B2$ , đến dòng thứ 2 là  $(A3*B3)+(A4*B4)$  và đến dòng thứ 3 là  $(A2*B2)+(A3*B3)+(A4*B4)$ . EXCEL sẽ thực hiện tính từng dòng trong CSDL theo công thức chỉ định và tích lũy lại theo phép tổng (SUM) và ra kết quả.

Nếu không dùng tổ hợp phím CTRL+SHIFT+ENTER thì giá trị ở cell sẽ là #VALUE! vì sai kiểu.

Trên chính là phân tích về Formula Array mà EXCEL thực hiện.

Từ đây chúng ta sẽ tự lập cho mình những công thức tính có kết hợp nhiều điều kiện, trong khi bạn lập bạn nên có cách tư duy như: tính tổng vùng TIEN nếu như NGÀY là Thứ 7, công thức sẽ là SUM(IF(WEEKDAY(NGÀY)=7,TIEN,0)).

Với công thức trên EXCEL sẽ cộng số TIEN ở dòng tương ứng là Thứ 7, nếu không phải cộng 0. Bạn cứ hình dung trình tự khi EXCEL phân tích công thức trên từng dòng và tích lũy lại theo phép toán tùy vào bạn dùng hàm gì bạn sẽ hiểu.

Trong VD trên là tổng vì dùng hàm SUM bên ngoài.

### **Phân tích cách dùng SumProduct và Công thức mảng.**

Hàm SumProduct:

Cấu trúc SUMPRODUCT(array1,array2,array3, ...)

Array - Mảng dữ liệu là một tập hợp dãy giá trị liên tiếp trong một hàng nào đó. VD A1:C1 hoặc A1:A10,...

Phép tính này cho phép chúng ta tính tổng của tích  $array1*array2*array3* \dots array30$ .

VD: A: Số lượng; B: Đơn giá  
A1 = 2 B1=20 C1="Cam" D1="Giống lai"  
A2 = 3 B2=10 C2="Bưởi" D2="Không"  
A3 = 4 B3=25 C3="Cam" D3="Không"

Bây giờ cần tính doanh thu của các loại hoa quả

array1=A1:A3

array2=B1:B3

Công thức =SumProduct(A1:A3, B1:B3) = 170

Bản chất công thức làm việc như thế này =A1\*B1+A2\*B2+A3\*B3 kết quả là 170

### **Nhắc lại về phép tính logic:**

Giá trị kiểu logic chỉ cho ra 1 trong 2 giá trị là TRUE/1, FALSE/0

Phép toán logic: <, >, <>, =, >=, <=, Not()

VD:

2>3=False

3>1=True

4>3=True

\*) Logic và - AND

=(2>3)\*(3>1)\*(4>3)=False\*True\*True=0\*1\*1=False/0 tương đương với hàm AND(2>3,3>1,4>3). Ít nhất một logic=False thì kết quả sẽ là False hay 0.

\* Logic hoặc - OR

=(2>3)+(3>1)+(4>3)=False+True+True=0+1=True/1 tương đương với hàm OR(2>3,3>1,4>3). Ít nhất một logic=True thì kết quả sẽ là True hay 1.

Lưu ý tổng của các giá trị là True=True=1).

### **\*) Tính tổng có nhiều điều kiện:**

Cách 1: dùng SUMPRODUCT

Tính tổng doanh thu của loại là "Cam"

=SUMPRODUCT(A1:A3,B1:B3\*(C1:C3="Cam")) hoặc

=SUMPRODUCT(A1:A3\*B1:B3\*(C1:C3="Cam")) = 140

Công thức tính như sau:

=A1\*B1\*(C1="Cam")+A2\*B2\*(C2="Cam")+A3\*B3\*(C3="Cam" )

=2\*20\*True+3\*10\*False+4\*25\*True

=2\*20\*1+3\*10\*0+4\*25\*1= 140

Cách 2: dùng Công thức mảng - "Formula Array"

=Sum(IF(C1:C3="Cam",A1:A3\*B1:B3,0))

Kết thúc nhấn tổ hợp CTRL+SHIFT+ENTER. Với cách làm này EXCEL sẽ phân tích như sau:

Xét trên từng dòng trong mảng (array)

dòng1: (c1="Cam")=true nên lấy A1\*B1=2\*20

dòng2: (c2="Cam")=false nên lấy 0 (theo cách của lấy của hàm IF)

dòng3: (c3="Cam")=true nên lấy A3\*B3=4\*25

Sau khi chạy hết các dòng, EXCEL sẽ dùng hàm SUM để tính tổng kết quả tính được ở từng dòng=2\*20+0+2\*25=140. Nếu trong công thức là hàm khác hàm SUM thì cách tính sẽ theo hàm đó.

### **Như vậy có 2 cách tính:**

=SUMPRODUCT(A1:A3\*B1:B3\*(C1:C3="Cam")) và  
=Sum(IF(C1:C3="Cam",A1:A3\*B1:B3,0))

**\*) Vậy tại sao không dùng là =SUM(A1:A3\*B1:B3\*(C1:C3="Cam"))  
mà phải dùng hàm =SUMPRODUCT(A1:A3\*B1:B3\*(C1:C3="Cam")) ?**

Cấu trúc của SUM là

SUM(number1,number2, ...)

Còn SUMPRODUCT là

SUMPRODUCT(array1,array2,array3, ...)

number <> array

Nếu SUMPRODUCT(A1:A3\*B1:B3\*(C1:C3="Cam")) rồi ENTER là đúng vì đối số của nó phải là mảng - Array.

Nếu công thức =SUM(A1:A3\*B1:B3\*(C1:C3="Cam")) rồi ENTER kết quả là #VALUE! -lỗi vì A1:A3 là một array chứ không phải là một number.

Nếu nhấn tổ hợp CTRL+SHIFT+ENTER. Với cách làm này EXCEL sẽ phân tích như sau:

Xét trên từng dòng trong mảng (array)

dòng1: A1\*B1\*(c1="Cam")=2\*20\*True=2\*20\*1

dòng2: A2\*B2\*(c2="Cam")=3\*10\*False=3\*10\*0

dòng3: A3\*B3\*(c3="Cam")=2\*25\*True=4\*25\*1

Sau khi chạy hết các dòng, EXCEL sẽ dùng hàm SUM để tính tổng kết quả tính được ở từng dòng=2\*20\*1+3\*10\*0  
+4\*25\*1=140.

**Vậy vẫn dùng được =SUM(A1:A3\*B1:B3\*(C1:C3="Cam")) với điều kiện nhấn tổ hợp phím CTRL+SHIFT+ENTER**

Như vậy đến đây chúng ta có 3 cách tính:

=SUMPRODUCT(A1:A3\*B1:B3\*(C1:C3="Cam")) nhấn phím ENTER

=SUM(IF(C1:C3="Cam",A1:A3\*B1:B3,0)) nhấn phím CTRL+SHIFT+ENTER

=SUM(A1:A3\*B1:B3\*(C1:C3="Cam")) nhấn phím CTRL+SHIFT+ENTER

Chúng có thể kết hợp rất nhiều điều kiện vào trong hàm thông qua phép toán logic nhân-và-And, cộng - hoặc - Or.

\*) Dùng hàm SUMPRODUCT hay dùng SUM kết hợp CTRL+SHIFT+ENTER đều cho ra được kết quả như nhau chính là do phép toán logic của bạn.

\*) Hàm SUMPRODUCT chỉ có thể tính tổng theo nhiều điều kiện

\*) Công thức mảng - Formula Array ngoài việc tính tổng có nhiều điều kiện còn làm rất nhiều phép tính khác do cách sử dụng hàm mà thôi.

Khi sử dụng công thức mảng SUM(IF())

## **TÓM TẮT**

Bạn có thể sử dụng chức năng bảng SUMIF() trong Microsoft Excel cho một trong những trường hợp sau:

- Khi bạn muốn trả lại một số tiền cho một tiêu chuẩn duy nhất (ví dụ: một máy nếu kiện)
- Khi bạn muốn sử dụng nhiều tiêu chí lại nhiều ô tổng

Tiêu chí bạn có thể sử dụng chức năng bảng SUMIF() giới hạn của văn bản, số hoặc dải và chức năng không thể sử dụng hằng số mảng.

Để biết thêm thông tin về hằng số mảng và chức năng bảng tính, hãy bấm vào số bài viết sau để xem bài viết trong cơ sở kiến thức Microsoft:

Bạn không thể sử dụng chức năng bảng SUMIF() logic toán tử Boolean, chẳng hạn như **hoặc** và **và**.

Làm thế nào để sử dụng một hợp lý và hoặc hoặc trong một tuyên bố SUM + nếu trong Excel

Khi bạn muốn sử dụng nhiều điều kiện, mảng hằng, lồng nhau nếu câu hoặc nhà khai thác Boolean; hoặc bạn muốn có một tổng số cho nhiều điều kiện (tiêu chuẩn); sử dụng hai chức năng SUM và nếu với nhau. Ví dụ, sử dụng SUM(IF()) thay vì các chức năng SUMIF().

Bài viết này chứng tỏ làm thế nào để sử dụng hai chức năng SUM và nếu với nhau thay vì chức năng bảng tính SUMIF().

Để sử dụng các khoản tiền và nếu chức năng bảng tính với nhau để tổng nhiều tiêu chí, tham khảo các ví dụ sau.

Ví dụ 1: Lồng nhau nếu chức năng với Boolean hoặc (+)

1. Bắt đầu Excel và tạo bảng tính sau:
2. A1: Dept B1: Employees C1: Criteria
3. A2: A B2: 2 C2: A
4. A3: B B3: 4 C3: B
5. A4: C B4: 3
6. A5: A B5: 3
7. A6: B B6: 3
8. A7: C B7: 2
9. A8: A B8: 4
10. A9: C B9: 3

11. Trong tế bào D1, gõ công thức sau đây:

**=SUM(IF((A2:A9="A")+(A2:A9="B"),B2:B9,0))**

12. Bấm CTRL + SHIFT + ENTER (hoặc COMMAND + lợi nhuận trên máy Mac) vào công thức như là một công thức mảng.

Công thức trả về 16, tổng số nhân viên trong phòng ban a và B.

Ví dụ 2: Lồng nhau nếu hoạt động với tiêu chí Array ({"A", "B"})

1. Lặp lại bước 1 ví dụ 1.
2. Trong tế bào D2, gõ công thức sau đây:  
**=SUM(IF(A2:A9={"A","B"},B2:B9,0))**
3. Bấm CTRL + SHIFT + ENTER (hoặc COMMAND + lợi nhuận trên máy Mac) vào công thức như là một công thức mảng.

Công thức một lần nữa trở về 16.

### Ví dụ 3: Chức năng SUMIF() Worksheet

1. Lặp lại bước 1 ví dụ 1.
2. Trong tế bào D3, gõ công thức sau đây:  
**=SUMIF(A2:A9,C2:C3,B2:B9)**
3. Nhấn ENTER (hoặc quay trở lại trên máy Mac) vào công thức. Công thức trả về 9, số lượng nhân viên đáp ứng tiêu chí A.
4. Lấy tay cầm điện và điền vào công thức xuống đến tế bào D4.

Công thức trả về 7, số lượng nhân viên đáp ứng tiêu chí sinh

Đối với Excel 2007 trở về sau thì việc thay thế công thức mảng đã được MS Office đưa ra hàm Sumifs . S cuối ta có thể hiểu là LỌC NHIỀU ĐIỀU KIỆN

## HÀM SUMIFS

Hàm SUMIFS() cho phép bạn tính tổng các ô trong vùng thỏa mãn 1 hoặc nhiều điều kiện cho trước. Trường hợp cho bài viết này là tính tổng tiền thâm niên của các nhân viên với điều kiện nhân viên đó làm trên 2 năm kinh nghiệm và phải là nam giới.

### Cú pháp:

**=SUMIFS** (sum\_range, criteria\_range1, criteria1, [criteria\_range2, criteria2], ...)

Trong đó :

- **sum\_range**: bắt buộc, là một vùng các ô để tính tổng, ví dụ A1:A10. Các ô có giá trị chuỗi hoặc rỗng thì được bỏ qua.
- **criteria\_range1**: bắt buộc, là một vùng các ô để so sánh với điều kiện tính tổng, ví dụ B2:B6
- **criteria1**: bắt buộc, là điều kiện để tính tổng, giá trị của criterial có thể là số, biểu thức, cột tham chiếu, chuỗi. Ví dụ: 45, ">45", "C3", "Góc Kinh Nghiệm" hoặc "45"
- **criteria\_range2, criteria2, ...** là tùy chọn, không bắt buộc, có hay không cũng không sao.

Excel cho phép tối đa là 127 cặp điều kiện criteria\_range/criteria.

Ở bài viết này, Góc Kinh Nghiệm sử dụng lại dữ liệu của bài post SUMIF(), và thêm một cột giới tính.

Dữ liệu như sau:

	A	B	C	D	E
1	<b>Họ tên</b>	<b>Giới tính</b>	<b>Số năm thâm niên</b>	<b>Tiền thâm niên</b>	
2	Nguyễn Văn A	Nam	3.5	17,500,000	
3	Trần Văn B	Nam	2.5	12,500,000	
4	Đoàn Thị C	Nữ	3.5	17,500,000	
5	Nguyễn Văn D	Nam	3.5	17,500,000	
6	Nguyễn Thị E	Nữ	1.5	7,500,000	
7	Tổng số thâm niên		<b>14.5</b>	<b>47,500,000</b>	
8	Số tiền thâm niên cho mỗi năm		<b>5,000,000</b>		

Như bạn đã thấy ở ô D7, kết quả là 47,500,000 là tổng tiền thâm niên cho nhân viên có trên 2 năm kinh nghiệm và giới tính là nam. Công thức tại ô D7 như sau:

**=SUMIFS(D2:D6,C2:C6,">2",B2:B6,"Nam")**